

MEDIDAS REPETIDAS NO TEMPO ANÁLISE USANDO O PROC MIXED

Euclides Braga MALHEIROS*

O PROC MIXED é um procedimento utilizado para modelos mistos que é uma generalização do modelo linear geral, separando no modelo os efeitos fixos dos aleatórios e é escrito como: $y = X\beta + Zv + \epsilon$ onde β é o vetor dos parâmetros associados aos efeitos fixos, v aos efeitos aleatórios, e ϵ vetor de erros aleatórios, sendo v e ϵ não correlacionados, com esperanças nulas e matrizes de covariâncias G e R , respectivamente.

O PROC MIXED permite informar a estrutura da matriz (G), através do comando RANDOM, e a dos erros (R), através do comando REPEATED.

Para este tipo de análise os dados devem estar na forma univariada.

A sintaxe do PROC MIXED é:

```
PROC MIXED <opções1>;  
CLASS <var. de classif.>;  
MODEL <var. dep.>=<efeitos fixos> / <opções2>;  
RANDOM <efeitos aleatórios em G> / <opções3>;  
REPEATED <efeito repetido> / <opções4>;  
MAKE "<Tabela>" OUT=<SDS>;  
RUN;
```

Algumas das <opções 1> são:

- **DATA**=<SDS> - especifica o SAS-DATA-SET a ser usado.
- **METHOD**=<ML|REML|MIVQUE0> - especifica o método a ser usado para estimar os componentes da variância.

Algumas das <opções 2> são:

- **HTYPE** =<n> - especifica o tipo da soma de quadrados.

O comando RANDOM especifica os efeitos aleatórios do modelo

Uma das possíveis <opções3> é:

- **TYPE** =<CS|AR(1)|SIMPLE|UN|...> - especifica a estrutura da matriz G (dos efeitos aleatórios), dentro de uma lista de opções.

O comando REPEATED especifica a estrutura de erros. Não tem nada a ver com o REPEATED do PROC GLM.

Uma das possíveis <opções4> é:

- **TYPE** =<CS|AR(1)|SIMPLE|UN|...> - especifica a estrutura da matriz E (dos erros), dentro de uma lista de opções.

* Departamento de Ciências Exatas – FCAV/UNESP, Campus de Jaboticabal. 14870-000 Jaboticabal SP

- **SUB=<efeito>** - especifica o efeito que identifica a unidade experimental. É assumida completa independência entre tais unidades, de tal forma que este comando produz uma estrutura bloco-diagonal em R com blocos idênticos.
- **R** – solicita a impressão da primeira matriz bloco-diagonal de R.
- **RCORR** - solicita a impressão da matriz R.

O comando MAKE é usado para criar arquivos a partir de Tabelas do OUTPUT.

Algumas das *Tabelas* do comando MAKE são:

- **FITTINGS** - Tabela das estatísticas usadas na seleção do modelo.
- **ML|REML|MIVQUE0** - Tabela das estimativas dos componentes da variância.
- **TESTS** – Testes de hipóteses associados aos efeitos fixos.
- **Etc.**

Os comandos **CONTRAST**, **ESTIMATE** e **LSMEANS** podem ser usadas da mesma forma do PROC GLM.

Como se viu até aqui a análise de medidas repetidas no tempo requer especial atenção na estrutura da matriz de variâncias e covariâncias.

A análise de dados com medidas repetidas pelo PROC MIXED é feita em dois passos, ou sejam:

- 1) Avaliar a estrutura da matriz de covariâncias.
- 2) Analisar a tendência dos tratamentos ao longo dos tempos.

Passo 1: Avaliação da estrutura da matriz de covariâncias para os dados do Exemplo 2.

Várias estruturas disponíveis no SAS podem ser avaliadas. Como a apresentação aqui tem um objetivo didático, apenas 3 estruturas serão estudadas, ou sejam: Composta simétrica – CS; Auto regressiva de 1ª ordem – AR(1), Desestruturada – UN.

Ou sejam:

Composta Simétrica - CS	Auto-Regr. 1ª ordem –AR(1)	Desestruturada – UN
$\begin{bmatrix} \sigma^2 + \sigma_1 & \sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma^2 + \sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1 & \sigma^2 + \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1 & \sigma_1 & \sigma^2 + \sigma_1 \end{bmatrix}$	$\begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix}$	$\begin{bmatrix} \sigma_1^2 & \sigma_{12} & \sigma_{13} & \sigma_{14} \\ \sigma_{21} & \sigma_2^2 & \sigma_{23} & \sigma_{24} \\ \sigma_{31} & \sigma_{32} & \sigma_3^2 & \sigma_{34} \\ \sigma_{41} & \sigma_{42} & \sigma_{43} & \sigma_4^2 \end{bmatrix}$

É interessante observar que o número de parâmetros a serem estimados depende da estrutura.

Para t tempos, na CS o número de parâmetros é 2, na AR(1) é 2 e na UN é $t(t+1)/2$.

Exercício 7. Verificar, usando o PROC MIXED, quais dessas três estruturas melhor se ajusta aos dados do Exemplo 2.

Programa para: Ler os dados na forma multivariada, criar um univariado, representar graficamente e testar a esfericidade da matriz Sigma (Matriz de variâncias e covariâncias entre os tempos).

```

/* ANÁLISE DOS DADOS DA TABELA 4*/
OPTIONS LS=78 PS=64;
/* CRIAÇÃO DOS SDS MULTI E UNIVARIADOS */
DATA MULTI;
INFILE "A:\MRT2M.TXT";
INPUT GN BL T1-T7;
PROC PRINT;
RUN;

DATA UNI (KEEP=GN BL TP Y);
SET MULTI;
TP=88; Y=T1; OUTPUT UNI;
TP=104; Y=T2; OUTPUT UNI;
TP=120; Y=T3; OUTPUT UNI;
TP=137.5; Y=T4; OUTPUT UNI;
TP=153.5; Y=T5; OUTPUT UNI;
TP=181.5; Y=T6; OUTPUT UNI;
TP=209.5; Y=T7; OUTPUT UNI;
RUN;
PROC PRINT DATA=MULTI;
PROC PRINT DATA=UNI;
RUN;

/* CRIANDO UM SDS UNI E MULTI - PERMANENTES NO DRIVE A*/
LIBNAME PASTA "A:\";
DATA PASTA.UNI; SET UNI;
DATA PASTA.MULTI; SET MULTI;
RUN;

/* REPRESENTAÇÃO GRÁFICA */
PROC SORT DATA=UNI; BY GN TP;
PROC MEANS NOPRINT;
OUTPUT OUT=AG MEAN=YG;
VAR Y;
BY GN TP;
RUN;
PROC GPLOT DATA=AG;
PLOT YG*TP=GN/GRID;
SYMBOL1 COLOR=RED INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL2 COLOR=BLUE INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL3 COLOR=GREEN INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL4 COLOR=BLACK INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL5 COLOR=ORANGE INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
TITLE HEIGHT=1.4 C=BLUE "PERFIL DOS GENOTIPOS AO LONGO DO TEMPO";
RUN;

/* Análise usando o comando REPEATED do PROC GLM */
PROC GLM DATA=MULTI;
CLASS BL GN;
MODEL T1-T7=BL GN/NOUNI;
REPEATED TP 6 POLYNOMIAL/PRINTE SUMMARY;
RUN;

/* Análise usando o PROC MIXED */
PROC MIXED DATA=UNI;
CLASS GN BL TP;
MODEL Y=BL GN TP GN*TP;
REPEATED TP/TYPE=AR(1) SUB=GN*BL R RCORR;
RUN;

```

Programa usando o PROC MIXED:

```
/* Análise usando o PROC MIXED */  
PROC MIXED DATA=UNI;  
CLASS GN BL TP;  
MODEL Y=BL GN TP GN*TP;  
REPEATED TP/TYPE=UN SUB=GN*BL R RCORR;  
RUN;
```

Dentre as diversas informações do output, a de interesse para a seleção da estrutura é a tabela dos critérios:

a) Para CS

Fit Statistics		
-2 Res Log Likelihood		322.6
AIC (smaller is better)		326.6
AICC (smaller is better)		326.7
BIC (smaller is better)		328.6
Null Model Likelihood Ratio Test		
DF	Chi-Square	Pr > ChiSq
1	0.43	0.5132

b) Para AR(1)

-2 Res Log Likelihood		315.1
AIC (smaller is better)		319.1
AICC (smaller is better)		319.3
BIC (smaller is better)		321.1
Null Model Likelihood Ratio Test		
DF	Chi-Square	Pr > ChiSq
1	7.87	0.0050

c) Para UN

-2 Res Log Likelihood		208.2
AIC (smaller is better)		264.2
AICC (smaller is better)		286.5
BIC (smaller is better)		292.1
Null Model Likelihood Ratio Test		
DF	Chi-Square	Pr > ChiSq
27	114.80	<.0001

Nessa parte do output são observadas várias estatísticas usadas para a seleção da melhor estrutura da matriz de covariâncias, ou sejam:

RLL – Res Log Likelihood - RLL,

AIC – Akaike's Information Criterion e

BIC – Schwarz's Bayesian Criterion.

Os critérios AIC e BIC são ajustes do RLL e são os mais usados na literatura.

Quanto menor o valor dessas estatísticas, melhor a estrutura.

Apresenta também: **Null Model Likelihood Ratio Test** que é uma estatística Qui-quadrado com respectivo graus de liberdade e probabilidade, para testar se a estrutura $\mathbf{I}\sigma^2$ (Usada no PROC GLM) é melhor que a especificada no PROC MIXED.

Nos exemplos, esses resultados podem ser vistos na seguinte Tabela:

Tabela – 5 Estatísticas utilizadas para a escolha da melhor estrutura para a matriz de variâncias e covariâncias dos tempos.

Estrutura	χ^2	GL	p-value	RLL	AIC	BIC
CS	0,43	1	0,5132	322,6	326,7	328,6
AR(1)	7,87	1	0,005	315.1	319.1	321.1
UN	114,80	27	<0,0001	208.2	264.2	292.1

A partir desta Tabela podemos concluir que:

Ao nível de 5% de probabilidade, a estrutura $\mathbf{I}\sigma^2$, usada no PROC GLM, é mais apropriada que a CS, o que não acontece com as estruturas AR(1) e UN.

Dentre as três estruturas avaliadas, a desestruturada (UN) é a mais apropriada, independente do critério utilizado.

A tabela dos critérios é especificada pelo comando MAKE como FITTING.

Como o programa precisa ser rodado com várias estruturas, e o que interessa são apenas as estatísticas utilizadas na seleção da estrutura, o programa deve salvar essas tabelas separadamente (comando MAKE):

Programa:

```
/* Análise usando o PROC MIXED */
LIBNAME PASTA "A:\";
PROC MIXED DATA=PASTA.UNI;
CLASS GN BL TP;
MODEL Y=BL GN TP GN*TP;
REPEATED TP/TYPE=CS SUB=GN*BL R RCORR;
MAKE "FITTING" OUT=SDS_CS;
RUN;
PROC PRINT DATA=SDS_CS;
RUN;
```

Arquivo SDS_CS:

Obs	Descr	Value
1	-2 Res Log Likelihood	322.6
2	AIC (smaller is better)	326.6
3	AICC (smaller is better)	326.7
4	BIC (smaller is better)	328.6

Observe que o nome da coluna Value deve ser renomeado, para que seja criada uma única tabela com esses valores. Para isso, trocar a linha MAKE por:

```
MAKE "FITTING" OUT=SDS_CS (RENAME=(VALUE=VAL_CS)) ;
```

Uma sugestão de programa completo usando Macro Subprograma seria:

```
/* Análise para escolha da estrutura - usando o PROC MIXED */
%MACRO E_SIGMA(SDSE, EST, SDSS, VAL);
PROC MIXED DATA=&SDSE;
CLASS GN BL TP;
MODEL Y=BL GN TP GN*TP;
REPEATED TP/TYPE=&EST SUB=GN*BL R RCORR;
MAKE "FITTING" OUT=&SDSS (RENAME= (VALUE=&VAL) );
RUN;
%MEND E_SIGMA;

LIBNAME SANTOS "A:\";
DATA A; SET SANTOS.UNI;
PROC PRINT DATA=A; RUN;
%E_SIGMA(A, CS, SDS_CS, V_CS);
%E_SIGMA(A, AR(1), SDS_AR, V_AR);
%E_SIGMA(A, UN, SDS_UN, V_UN);
RUN;
PROC PRINT DATA=SDS_CS;
PROC PRINT DATA=SDS_AR;
PROC PRINT DATA=SDS_UN;
RUN;

DATA RESULT; MERGE SDS_CS SDS_AR SDS_UN;
RUN;
PROC PRINT DATA=RESULT;
RUN;
```

Resultados:

	Descr	V_CS	V_AR	V_UN
1	-2 Res Log Likelihood	322.6	315.1	208.2
2	AIC (smaller is better)	326.6	319.1	264.2
3	AICC (smaller is better)	326.7	319.3	286.5
4	BIC (smaller is better)	328.6	321.1	292.1

Analisando os resultados conclui-se que a estrutura mais adequada, entre as estudadas, é a desestruturada (UN).

Esse programa pode ser usado testando todas as estruturas disponíveis no SAS. Algumas das opções são:

Estrutura	Descrição	Parâmetros
ANTE(1)	Ante-Dependence	2t-1
AR(1)	Autoregressive	2
ARH(1)	Heterogeneous Autoregressive	t+1
ARMA(1,1)	ARMA(1,1)	3
CS	Compound Symmetric	2
CSH	Heterogeneous Compound Symmetric	t+1
FA(q)	Factor Analytic	q/2(2t-q+1)+t
FA0(q)	No Factor Analytic	q/2(2t-q+1)
HF	Huynh-Feldt	t+1
LIN(q)	General Linear	q
TOEP	Toeplitz	t
TOEP(q)	Banded Toeplitz	q
TOEPH	Heterogeneous Toeplitz	2t-1
TOEPH(q)	Banded Heterogeneous Toeplitz	t+q-1
UN	Unstructured	t(t+1)/2
UNAR(q)	Banded	q/2(2t-q+1)
UNR	Unstructured Corrs	t(t+1)/2
VC	Variance Components	q

<p>Auto-Regr. 1ª ordem –AR(1)</p> $\sigma^2 \begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix}$	<p>Auto-Regr. Harm. –ARH(1)</p> $\begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho & \sigma_1\sigma_3\rho^2 & \sigma_1\sigma_4\rho^3 \\ \sigma_2\sigma_1\rho & \sigma_2^2 & \sigma_2\sigma_3\rho & \sigma_2\sigma_4\rho^2 \\ \sigma_3\sigma_1\rho^2 & \sigma_3\sigma_2\rho & \sigma_3^2 & \sigma_3\sigma_4\rho \\ \sigma_4\sigma_1\rho^3 & \sigma_4\sigma_2\rho^2 & \sigma_4\sigma_3\rho & \sigma_4^2 \end{bmatrix}$
<p>Composta Simétrica - CS</p> $\begin{bmatrix} \sigma^2 + \sigma_1 & \sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma^2 + \sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1 & \sigma^2 + \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1 & \sigma_1 & \sigma^2 + \sigma_1 \end{bmatrix}$	<p>Composta Sim. Harm – CSH</p> $\begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho & \sigma_1\sigma_3\rho^2 & \sigma_1\sigma_4\rho^3 \\ \sigma_2\sigma_1\rho & \sigma_2^2 & \sigma_2\sigma_3\rho & \sigma_2\sigma_4\rho^2 \\ \sigma_3\sigma_1\rho^2 & \sigma_3\sigma_2\rho & \sigma_3^2 & \sigma_3\sigma_4\rho \\ \sigma_4\sigma_1\rho^3 & \sigma_4\sigma_2\rho^2 & \sigma_4\sigma_3\rho & \sigma_4^2 \end{bmatrix}$
<p>TOEP</p> $\begin{bmatrix} \sigma^2 & \sigma_1 & \sigma_2 & \sigma_3 \\ \sigma_1 & \sigma_1^2 & \sigma_1 & \sigma_2 \\ \sigma_2 & \sigma_1 & \sigma^2 & \sigma_1 \\ \sigma_3 & \sigma_2 & \sigma_1 & \sigma^2 \end{bmatrix}$	<p>TOEPH</p> $\begin{bmatrix} \sigma_1^2 & \sigma_1\sigma_2\rho_1 & \sigma_1\sigma_3\rho_2 & \sigma_1\sigma_4\rho_3 \\ \sigma_2\sigma_1\rho_1 & \sigma_2^2 & \sigma_2\sigma_3\rho_1 & \sigma_2\sigma_4\rho_3 \\ \sigma_3\sigma_1\rho_2 & \sigma_3\sigma_2\rho_1 & \sigma_3^2 & \sigma_3\sigma_4\rho_1 \\ \sigma_4\sigma_1\rho_3 & \sigma_4\sigma_2\rho_2 & \sigma_4\sigma_3\rho_1 & \sigma_4^2 \end{bmatrix}$
<p>Desestruturada – UN</p> $\begin{bmatrix} \sigma_1^2 & \sigma_{12} & \sigma_{13} & \sigma_{14} \\ \sigma_{21} & \sigma_2^2 & \sigma_{23} & \sigma_{24} \\ \sigma_{31} & \sigma_{32} & \sigma_3^2 & \sigma_{34} \\ \sigma_{41} & \sigma_{42} & \sigma_{43} & \sigma_4^2 \end{bmatrix}$	<p>Desestruturada – UN(1)</p> $\begin{bmatrix} \sigma_1^2 & 0 & 0 & 0 \\ 0 & \sigma_2^2 & 0 & 0 \\ 0 & 0 & \sigma_3^2 & 0 \\ 0 & 0 & 0 & \sigma_4^2 \end{bmatrix}$

Programa:

```
/* Análise para escolha da estrutura - usando o PROC MIXED */
%MACRO E_SIGMA(SDSE,EST,SDSS,VAL);
PROC MIXED DATA=&SDSE;
CLASS GN BL TP;
MODEL Y=BL GN TP GN*TP;
REPEATED TP/TYPE=&EST SUB=GN*BL R RCORR;
MAKE "FITTING" OUT=&SDSS(RENAME=(VALUE=&VAL));
RUN;
%MEND E_SIGMA;

LIBNAME SANTOS "A:\";
DATA A; SET SANTOS.UNI;
PROC PRINT DATA=A; RUN;
%E_SIGMA(A, ANTE(1), SDS_ANTE, V_ANTE);
%E_SIGMA(A, AR(1), SDS_AR, V_AR);
%E_SIGMA(A, ARH(1), SDS_ARH, V_ARH);
%E_SIGMA(A, ARMA(1,1), SDS_ARMA, V_ARMA);
%E_SIGMA(A, CS, SDS_CS, V_CS);
%E_SIGMA(A, CSH, SDS_CSH, V_CSH);
%E_SIGMA(A, FA0(1), SDS_FA0, V_FA0);
%E_SIGMA(A, FA(1), SDS_FA1, V_FA1);
%E_SIGMA(A, HF, SDS_HF, V_HF);
%E_SIGMA(A, SIMPLE, SDS_SIMPLE, V_SIMPLE);
%E_SIGMA(A, TOEP, SDS_TOEP, V_TOEP);
%E_SIGMA(A, TOEPH, SDS_TOEPH, V_TOEPH);
%E_SIGMA(A, UN, SDS_UN, V_UN);
%E_SIGMA(A, UNAR, SDS_UNAR, V_UNAR);
%E_SIGMA(A, UNCS, SDS_UNCS, V_UNCS);
%E_SIGMA(A, UNR, SDS_UNR, V_UNR);
%E_SIGMA(A, VC, SDS_VC, V_VC);
RUN;
DATA RESULT;
MERGE SDS_ANTE SDS_AR SDS_ARH SDS_ARMA SDS_CS SDS_CSH SDS_FA0
      SDS_FA1 SDS_HF SDS_SIMPLE SDS_TOEP SDS_TOEPH
      SDS_UN SDS_UNAR SDS_UNCS SDS_UNR SDS_VC;
RUN;
PROC PRINT DATA=RESULT;
RUN;
```

Escolhida a estrutura incluir os comandos de interesse para as comparações múltiplas, contrastes, desdobramentos, etc.

```
/* ANÁLISE COM A ESTRUTURA ESCOLHIDA */

LIBNAME SANTOS "A:\";
DATA A; SET SANTOS.UNI;
PROC PRINT DATA=A;
RUN;
PROC MIXED DATA=A;
CLASS GN BL TP;
MODEL Y=BL GN TP GN*TP;
REPEATED TP/TYPE=UN SUB=GN*BL R RCORR;
```



```

*LSMEANS GN/PDIFF ADJUST=TUKEY;

*TESTAR H0:M_TR=M_CONT (CONT=TR1);
*LSMEANS GN/PDIFF=CONTROL ADJUST=DUNNETT;

*TESTAR H0:M_TR<M_CONT (CONT=TR1);
*LSMEANS GN/PDIFF=CONTROL ADJUST=DUNNETT;

*TESTAR H0:M_TR>M_CONT (CONT=TR1);
*LSMEANS GN/PDIFF=CONTROL ADJUST=DUNNETT;

*TESTAR H0:M_TR=M_CONT (CONT=TR2);
*LSMEANS GN/PDIFF=CONTROL('2') ADJUST=DUNNETT;

*DESDOBRAMENTO DA INTERAÇÃO;
*LSMEANS GN*TP/Slice=TP;
*LSMEANS GN*TP/Slice=GN;
/*
* DESDOBRAMENTO DOS GL DE GN, POR CONTRASTES;
CONTRAST " (G1+G2) vs (G3+G4+G5) " GN 3 3 -2 -2 -2;
CONTRAST "G1 vs G2" GN 1 -1 0 0 0;
CONTRAST "D. (G1 G2 G3) " GN 0 0 2 -1 -1,
                        GN 0 0 0 1 -1;

* DESDOBRAMENTO DOS GL DE TP, POR CONTRASTES DE HELMERT;
CONTRAST "1 VS POST." TP 6 -1 -1 -1 -1 -1 -1;
CONTRAST "2 VS POST." TP 0 5 -1 -1 -1 -1 -1;
CONTRAST "3 VS POST." TP 0 0 4 -1 -1 -1 -1;
CONTRAST "4 VS POST." TP 0 0 0 3 -1 -1 -1;
CONTRAST "5 VS POST." TP 0 0 0 0 2 -1 -1;
CONTRAST "6 VS POST." TP 0 0 0 0 0 1 -1;
*/

```

RUN;

Exemplo 4:

Considere os dados de um experimento de degradação ruminal instalado num delineamento em quadrado latino 6x6 – 6 períodos, 6 animais e 6 tratamentos, avaliado em 10 tempos (3, 6, 12, 24, 48, 60, 72, 84, 96 e 120h). Foram avaliadas três variáveis dependentes (Y1=DEG, Y2=FDN e Y3=FDA).

Os dados encontram-se no arquivo DQL.XLX.

Exercício 8. Fazer um programa SAS para:

- Criar os SDS necessários.
- Representar graficamente o perfil dos tratamentos ao longo do tempo.
- Testar a esfericidade da matriz de covariâncias, considerando animal e período como efeitos aleatórios.
- Analisar os dados usando modelos mistos, se for o caso.

Programa:

```

/* ANÁLISE PARA ESCOLHA DA ESTRUTURA - USANDO O PROC MIXED */
OPTIONS LS=78 PS=64 PAGENO=1;

* IMPORTANDO O ARQUIVO UNI (AU) - ARQUIVO ASC;
DATA AU;
INFILE "A:\DQL.TXT" FIRSTOBS=2;
INPUT TP PR AN TR Y1-Y3;
LABEL Y1="DEG" Y2="FDN" Y3="FDA";
PROC PRINT;
RUN;

%LET Y=Y1;
DATA AU; SET AU;
KEEP TP PR AN TR &Y;
PROC PRINT;
RUN;

* CRIANDO O SDS MULTI (AM);
PROC SORT DATA=AU; BY PR AN TR;
PROC TRANSPOSE OUT=AM(RENAME=( _3=T1 _6=T2 _12=T3 _24=T4 _48=T5
    _60=T6 _72=T7 _84=T8 _96=T9 _120=T10));
BY PR AN TR;
ID TP;
RUN;
PROC PRINT DATA=AM;
RUN;

*REPRESENTAÇÃO GRÁFICA;
PROC SORT DATA=AU; BY TR TP;
PROC MEANS NOPRINT;
OUTPUT OUT=AG MEAN=YG;
VAR &Y;
BY TR TP;
RUN;
PROC GPLOT DATA=AG;
PLOT YG*TP=TR/GRID;
SYMBOL1 COLOR=RED INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL2 COLOR=BLUE INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL3 COLOR=GREEN INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL4 COLOR=BLACK INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL5 COLOR=ORANGE INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
SYMBOL6 COLOR=CYAN INTERPOL=JOIN VALUE=DOT HEIGHT=0.5;
TITLE HEIGHT=1.4 C=BLUE "PERFIL DOS TRATAMENTOS AO LONGO DO TEMPO";
RUN;

* TESTE DE ESFERICIDADE DA MATRIZ SIGMA;
PROC GLM DATA=AM;
CLASS PR AN TR ;
MODEL T1-T10=PR AN TR/NOUNI;
REPEATED TP 6 (3 6 12 24 48 60 72 84 96 120) POLYNOMIAL/PRINTE SUMMARY;
RUN;

* ESCOLHA DA ESTRUTURA DE SIGMA;
%MACRO E_SIGMA(SDSE,EST,SDSS,VAL);
PROC MIXED DATA=&SDSE;
CLASS PR AN TR TP;

```

```

MODEL &Y=PR AN TR TP TR*TP/HTYPE=3 ;
RANDOM AN PR;
REPEATED TP/TYPE=&EST SUB=TR(PR AN) R RCORR;
MAKE "FitStatistics" OUT=&SDSS (RENAME=(VALUE=&VAL) );
RUN;
%MEND E_SIGMA;


DATA A; SET AU;
PROC PRINT DATA=A; RUN;
*%E_SIGMA(A, ANTE(1), SDS_ANTE, V_ANTE);
%E_SIGMA(A, AR(1), SDS_AR, V_AR);
*%E_SIGMA(A, ARH(1), SDS_ARH, V_ARH);
*%E_SIGMA(A, ARMA(1,1), SDS_ARMA, V_ARMA);
*%E_SIGMA(A, CS, SDS_CS, V_CS);
*%E_SIGMA(A, CSH, SDS_CSH, V_CSH);
*%E_SIGMA(A, FA0(1), SDS_FA0, V_FA0);
*%E_SIGMA(A, FA(1), SDS_FA1, V_FA1);
*%E_SIGMA(A, HF, SDS_HF, V_HF);
*%E_SIGMA(A, SIMPLE, SDS_SIMPLE, V_SIMPLE);
%E_SIGMA(A, TOEP, SDS_TOEP, V_TOEP);
*%E_SIGMA(A, TOEPH, SDS_TOEPH, V_TOEPH);
%E_SIGMA(A, UN, SDS_UN, V_UN);
*%E_SIGMA(A, UNAR, SDS_UNAR, V_UNAR);
*%E_SIGMA(A, UNCS, SDS_UNCS, V_UNCS);
*%E_SIGMA(A, UNR, SDS_UNR, V_UNR);
*%E_SIGMA(A, VC, SDS_VC, V_VC);
RUN;

DATA RESULT;
*MERGE SDS_ANTE SDS_AR SDS_ARH SDS_ARMA SDS_CS SDS_CSH SDS_FA0
      SDS_FA1 SDS_HF SDS_SIMPLE SDS_TOEP SDS_TOEPH
      SDS_UN SDS_UNAR SDS_UNCS SDS_UNR SDS_VC;
MERGE SDS_AR SDS_TOEP SDS_UN;
RUN;
PROC PRINT DATA=RESULT;
RUN;

```

Para a variável Y1 obteve-se:

a) Gráfico:



a) Estatísticas para escolha da estrutura:

Obs	Descr	V_ANTE	V_AR	V_ARH	V_ARMA	V_CS	V_CSH
1	-2 Res Log Likelihood	1360.8	1616.1	1417.4	1615.0	1617.8	1442.0
2	AIC (smaller is better)	1400.8	1620.1	1439.4	1621.0	1621.8	1464.0
3	AICC (smaller is better)	1403.9	1620.1	1440.3	1621.1	1621.9	1464.9
4	BIC (smaller is better)	1396.6	1619.7	1437.1	1620.4	1621.4	1461.7
		V_FA1	V_SIMPLE	V_TOEP	V_TOEPH	V_UN	V_UNR
		1362.4	1624.4	1600.6	1400.9	1307.0	1307.0
		1400.4	1626.4	1620.6	1438.9	1417.0	1417.0
		1403.2	1626.4	1621.4	1441.8	1443.5	1443.5
		1396.4	1626.2	1618.6	1435.0	1405.5	1405.5
						V_VC	
							1624.4
							1626.4
							1626.4
							1626.2

b) Observe que a melhor estrutura é a UN (desestruturada) e que as estruturas HF e UNCS apresentaram problemas na convergência do método iterativo.

Exercício 8: Analisar os dados dos Exemplos 1 (DIC) , 2 (DBC) e 4(DQL), usando o **módulo analyst** do SAS.

Para esse tipo de análise os dados devem estar na forma “Univariada” Os arquivos EXEL correspondentes são: